

1. Mostrar cada uno de los nombres de los bailes con la etiqueta "losbailes".

```
<losbailes>{  
  for $b in /bailes/baile  
  return <baile>{$b/nombre/text()}</baile>  
}</losbailes>
```

2. Mostrar los nombres de los bailes seguidos con el número de plazas entre paréntesis, ambos dentro de la misma etiqueta "losbailes".

```
<losbailes>{  
  for $b in /bailes/baile  
  return <baile>{concat($b/nombre, " (", $b/plazas, ")")}</baile>  
}</losbailes>
```

3. Mostrar los nombres de los bailes cuyo precio sea mayor de 30.

```
for $b in /bailes/baile  
where xs:decimal($b/precio) > 30  
return $b/nombre
```

4. Mostrar los nombres de los bailes cuyo precio sea mayor de 30 y la moneda "euro".

```
for $b in /bailes/baile  
where xs:decimal($b/precio) > 30 and $b/precio/@moneda = "euro"  
return $b/nombre
```

5. Mostrar los nombres y la fecha de comienzo de los bailes que comiencen el mes de enero (utiliza para buscarlo la cadena de texto "/1/").

```
for $b in /bailes/baile  
where contains($b/comienzo, "/1/")  
return <baile>  
  <nombre>{$b/nombre/text()}</nombre>  
  <comienzo>{$b/comienzo/text()}</comienzo>  
</baile>
```

6. Mostrar los nombres de los profesores y la sala en la que dan clase, ordénalos por sala.

```
for $b in /bailes/baile  
order by $b/sala  
return <profesor>  
  <nombre>{$b/profesor/text()}</nombre>  
  <sala>{$b/sala/text()}</sala>  
</profesor>
```

7. Mostrar los nombres de los profesores eliminando los repetidos y acompañar cada nombre con todas las salas en la que da clase, ordénalos por nombre.

```
for $prof in distinct-values(/bailes/baile/profesor)
let $salas := distinct-values(/bailes/baile[profesor = $prof]/sala)
order by $prof
return <profesor>
  <nombre>{$prof}</nombre>
  <salas>{
    for $s in $salas
    return <sala>{$s}</sala>
  }</salas>
</profesor>
```

8. Mostrar la media de los precios de todos los bailes.

```
let $precios := /bailes/baile/precio
return <media>{avg(for $p in $precios return xs:decimal($p))}</media>
```

9. Mostrar la suma de los precios de los bailes de la sala 1.

```
let $precios := /bailes/baile[sala = 1]/precio
return <suma>{sum(for $p in $precios return xs:decimal($p))}</suma>
```

10. Mostrar cuántas plazas en total oferta el profesor "Jesus Lozano".

```
let $plazas := /bailes/baile[profesor = "Jesus Lozano"]/plazas
return <totalPlazas>{sum(for $p in $plazas return xs:integer($p))}</totalPlazas>
```

11. Mostrar el dinero que ganaría la profesora "Laura Mendiola" si se completaran todas las plazas de su baile, sabiendo que sólo tiene un baile.

```
let $b := /bailes/baile[profesor = "Laura Mendiola"]
return <ganancia>{
  xs:decimal($b/precio) * xs:integer($b/plazas)
}</ganancia>
```

12. Mostrar el dinero que ganaría el profesor "Jesus Lozano" si se completaran todas las plazas de su baile, pero mostrando el beneficio de cada baile por separado.

```
for $b in /bailes/baile[profesor = "Jesus Lozano"]
return <baile>
  <nombre>{$b/nombre/text()}</nombre>
  <ganancia>{xs:decimal($b/precio) * xs:integer($b/plazas)}</ganancia>
</baile>
```

13. Mostrar el dinero que ganaría la profesora "Laura" (no conocemos su apellido) si se completaran todas las plazas de su baile.

```
for $b in /bailes/baile[starts-with(profesor, "Laura")]
return <ganancia>{xs:decimal($b/precio) * xs:integer($b/plazas)}</ganancia>
```

14. Mostrar el nombre del baile, su precio y el precio con un descuento del 15% para familias numerosas. Ordenar por el nombre del baile

```
for $b in /bailes/baile
order by $b/nombre
return <baile>
  <nombre>{$b/nombre/text()}</nombre>
  <precioOriginal>{$b/precio/text()}</precioOriginal>
  <precioDescuento>{xs:decimal($b/precio) * 0.85}</precioDescuento>
</baile>
```

15. Mostrar todos los datos de cada baile excepto la fecha de comienzo y de fin.

```
for $b in /bailes/baile
return <baile id="{ $b/@id}">
  {$b/nombre}
  {$b/precio}
  {$b/plazas}
  {$b/profesor}
  {$b/sala}
</baile>
```

16. Mostrar en una tabla de HTML los nombres de los bailes y su profesor, cada uno en una fila.

```
<html>
<body>
  <table border="1">
    <tr>
      <th>Baile</th>
      <th>Profesor</th>
    </tr>
    {
      for $b in /bailes/baile
      return <tr>
        <td>{$b/nombre/text()}</td>
        <td>{$b/profesor/text()}</td>
      </tr>
    }
  </table>
</body>
</html>
```